Pensieve header: Implementing the scapegoated VdV algebra of http://drorbn.net/bbs/show?shot=-VanDerVeen-160731-121550.jpg.

```
ϵ = 0;
PBWBasis = {c, u, w};
B[U@c, U@w] = -(B[U@w, U@c] = U@w);
B[U@u, U@c] = -(B[U@c, U@u] = U@u);
B[U@w, U@u] = -(B[U@u, U@w] = (t - 1) U[](*-ϵ U[u,w]+ϵ t U[]+2ϵ t U[c]*));

Ui_[ℰ_] := ℰ /. {t → ti, u_U :> Replace[u, x_ :> xi, 1]};

B[x_, x_] = 0;
B[U[(x_)i_], U[(y_)i_]] := Ui[B[U@x, U@y]];
B[U[(x_)i_], U[(y_)j_]] /; i =!= j := 0;
B[x_, y_] := x ** y - y ** x;

x_ ≼ y_ := OrderedQ[{x, y}]; x_ ≺ y_ := ! OrderedQ[{y, x}];
Simp[ℰ_] := Collect[ℰ, _U, Expand];

Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
U[x_] ** U[y_] := (*U[x]**U[y] =*) If[x ≺ y, U[x, y], U[y, x] + B[U@x, U@y]];
U[x_] ** U[y1_, yy__] := (*U[x] ** U[y1,yy] =*)
  If[x ≼ y1, U[x, y1, yy], (U@x ** U@y1) ** U@yy];
U[xx__, xn_] ** U[yy__] := (*U[xx,xn]**U[yy] =*) U@xx ** (U@xn ** U@yy);

U[l___, x_^n_, r___] := U[l, Sequence @@ Table[x, {n}], r];
U[l___, 1, r___] := U[l, r];

LBasis[n_Integer] := LBasis[Range[n]];
LBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
       Table[{U@ci, U@ui, U@wi, ϵ U@ci, ϵ U@ui, ϵ U@wi}, {i, S}],
       Table[{U[ui, wj], ϵ U[ui, wj],
         ϵ U@@Sort@{ci, cj}, ϵ U[ci, uj], ϵ U[ci, wj]}, {i, S}, {j, S}],
       Table[{ϵ U[ci, uj, wk], ϵ U@@Sort@{ui, uj, wk}, ϵ U@@Sort@{ui, wj, wk}},
        {i, S}, {j, S}, {k, S}],
       Table[ϵ U@@Sort@{ui, uj, wk, wl}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]
```

```
BLBasis[n_Integer] := BLBasis[Range[n]];
BLBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
       Table[{U@cᵢ, ϵ U@cᵢ}, {i, S}],
       Table[{U[uᵢ, wⱼ], ϵ U[uᵢ, wⱼ], ϵ U@@Sort@{cᵢ, cⱼ}}, {i, S}, {j, S}],
       Table[{ϵ U[cᵢ, uⱼ, wₖ]}, {i, S}, {j, S}, {k, S}],
       Table[ϵ U@@Sort@{uᵢ, uⱼ, wₖ, wₗ}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]

UExp[ℰ_, n_] := Module[{t = U[], k}, U[] + Sum[t = t ** ℰ / k!, {k, n}]] // Simp

ToDegree[n_][ℰ_] :=
 Simp[ℰ] /. {ϵ → ℏ ϵ, bᵢ_ :> ℏ bᵢ, tᵢ_ :> ℯ^(ℏ bᵢ), b → ℏ b, t :> ℯ^(ℏ b), x_U :> ℏ^Count[x,u|u_] x} /.
  a_. x_U :> Series[a, {ℏ, 0, n}] * x
```

Testing AS and Jacobi

```
B[U@w, U@c]
```

```
U[w]
```

```
B[U@u, U@c]
```

```
-U[u]
```

```
B[U@w, U@u]
```

```
-(-1 + t) U[]
```

```
BLBasis[2]
```

```
{U[], U[c₁], U[c₂], U[u₁, w₁], U[u₁, w₂], U[u₂, w₁], U[u₂, w₂]}
```

```
bas = BLBasis[2]; Table[B[x, y] + B[y, x], {x, bas}, {y, bas}] // Flatten // Union
```

```
{0}
```

```
bas = BLBasis[2]; Timing[
 Table[
    {x, y, z} = xyz;
    Simp[B[B[x, y], z] + B[B[y, z], x] + B[B[z, x], y]],
    {xyz, Subsets[bas, {3}]}
   ] // Flatten // Union
]
```

```
{0.03125, {0}}
```

Testing Yang-Baxter

```
R[i_, j_, d_] := Module[{nn, p},
  Sum[
   Sum[nn = p - m;
    b_i^m / (m!) (nn!) U[u_i^nn] ** U[c_j^m] ** U[w_j^nn],
    {m, 0, p}],
   {p, 0, d}
  ]
 ]
```

```
Simp[R[1, 2, 3] /. x_U :> (t_1 - 1)/b_1)^Count[x, u_1] x] // ToDegree[3]
```

$$U[] + (b_1\, U[c_2] + U[u_1, w_2])\,\hbar + \left(\frac{1}{2} b_1^2\, U[c_2, c_2] + \frac{1}{2} b_1\, U[u_1, w_2] + b_1\, U[c_2, u_1, w_2] + \frac{1}{2} U[u_1, u_1, w_2, w_2]\right)\hbar^2 + \left(\frac{1}{6} b_1^2\, U[u_1, w_2] + \frac{1}{6} b_1^3\, U[c_2, c_2, c_2] + \frac{1}{2} b_1^2\, U[c_2, u_1, w_2] + \frac{1}{2} b_1^2\, U[c_2, c_2, u_1, w_2] + \frac{1}{2} b_1\, U[u_1, u_1, w_2, w_2] + \frac{1}{2} b_1\, U[c_2, u_1, u_1, w_2, w_2] + \frac{1}{6} U[u_1, u_1, u_1, w_2, w_2, w_2]\right)\hbar^3 + O[\hbar]^4$$

```
R3[d_] := (R[1, 2, d] ** R[1, 3, d] ** R[2, 3, d]) - (R[2, 3, d] ** R[1, 3, d] ** R[1, 2, d])
```

```
R3[1] // Simp // ToDegree[1]
```

$$O[\hbar]^3$$

```
R3[2] // ToDegree[2]
```

$$O[\hbar]^4$$

```
R3[2] // ToDegree[3]
```

$$O[\hbar]^4$$

```
R3[3] // ToDegree[4]
```

$$O[\hbar]^5$$